Universidad del País Vasco — Euskal Herriko Unibertsitatea

eman ta zabal zazu

Autoevaluación

OCW 2020: *Parametrización y representación gráfica de superficies construidas*

Tema 1. Ejercicios propuestos resueltos

Equipo docente del curso
Martín Yagüe, Luis
Barrallo Calonge, Javier
Soto Merino, Juan Carlos
Lecubarri Alonso, Inmaculada

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I (EIB/BIE)
ETS de Arquitectura de Donostia-San Sebastián (ETSASS/DAGET)

EJERCICIOS DEL TEMA 1. PARAMETRIZACIÓN DE CURVAS PLANAS

Ejercicio nº1

Enunciado

En el año 1989, el conocido arquitecto suizo Mario Botta construyó una vivienda unifamiliar en el municipio de Vacallo (Cantón del Tizino, Suiza). Esta vivienda, de planta triangular rectangular, presenta en su hipotenusa la fachada principal de composición simétrica. El motivo ornamental de la misma es una arquería formada por dos arcos entrelazados de medio punto estando los sálmeres al mismo nivel y resultando su canto formado por cinco roscas concéntricas de ladrillo cara-vista colocado a sardinel simple.

Se adjunta el siguiente enlace a la página *web* del estudio del arquitecto (*Mario Botta Architetti*) donde se muestra una fotografía de la fachada principal de la vivienda:

http://www.botta.ch/it/SPAZI DELL-ABITARE?idx=10

Parametrice las circunferencias que delimitan el área que conforman las testas de ladrillo del alzado de la arquería sabiendo que cada arco tiene un radio interior de 450 centímetros, su canto es de 130 centímetros y que la distancia entre los arcos es de 580 centímetros.

Represente gráficamente el área común a ambos arcos.



Imagen 1. Explicación de conceptos arquitectónicos (imagen propia)

Resolución

```
Remove["Global`*"]
```

- Por simplicidad, se consideran las medidas en metros
- En primer lugar, se establece un sistema cartesiano de referencia con origen en el centro de uno de los dos arcos; por ejemplo, el situado a la izquierda de la imagen que se denota como Arco 1
- Cada uno de los arcos (media corona circular) está delimitado por dos semicircunferencias
- Ecuaciones cartesianas de las curvas
  - Arco 1: delimitado por dos circunferencias, C_1 la interior y C_2 la exterior, centradas en el origen de coordenadas $O = (0, 0)$ siendo sus radios $R_1 = 4.5\,\mathrm{m}$ y $R_2 = 5.8\,\mathrm{m}$, respectivamente

$$C_1 \equiv x^2 + y^2 = 4.5^2 \quad \forall\ x \in [-4.5,\ 4.5]$$

$$C_2 \equiv x^2 + y^2 = 5.8^2 \quad \forall\ x \in [-5.8,\ 5.8]$$

- Arco 2 (el situado a la derecha de la imagen): delimitado por dos circunferencias, C_3 la interior y C_4 la exterior, centradas en el punto $A = (5.8,\ 0)$ siendo sus radios $R_3 = 4.5\ \mathrm{m}$ y $R_4 = 5.8\ \mathrm{m}$, respectivamente

$$C_3 \equiv (x - 5.8)^2 + y^2 = 4.5^2 \quad \forall\ x \in [1.3,\ 10.3]$$

$$C_4 \equiv (x - 5.8)^2 + y^2 = 5.8^2 \quad \forall\ x \in [0,\ 11.6]$$

- Ecuaciones paramétricas de las curvas:
  - Arco 1

$$\boxed{C_1 \equiv \begin{cases} x(t) = 4.5 \cos(t) \\ y(t) = 4.5 \operatorname{sen}(t) \end{cases} \quad \forall\ t \in [0,\ 2\pi]} \qquad \boxed{C_2 \equiv \begin{cases} x(t) = 5.8 \cos(t) \\ y(t) = 5.8 \operatorname{sen}(t) \end{cases} \quad \forall\ t \in [0,\ 2\pi]}$$

  - Arco 2

$$\boxed{C_3 \equiv \begin{cases} x(t) = & 5.8 + 4.5 \cos(t) \\ y(t) = & 4.5 \operatorname{sen}(t) \end{cases} \quad \forall\ t \in [0,\ 2\pi]} \qquad \boxed{C_4 \equiv \begin{cases} x(t) = & 5.8 + 5.8 \cos(t) \\ y(t) = & 5.8 \operatorname{sen}(t) \end{cases} \quad \forall\ t \in [0,\ 2\pi]}$$

- Representación gráfica
  - definición de ecuaciones

```
c1 = {x1 = 4.5 Cos[t], y1 = 4.5 Sin[t]};
c2 = {x2 = 5.8 Cos[t], y2 = 5.8 Sin[t]};

c3 = {x3 = 5.8 + 4.5 Cos[t], y3 = 4.5 Sin[t]};
c4 = {x4 = 5.8 + 5.8 Cos[t], y4 = 5.8 Sin[t]};
```

  - arquería

```
suelo = ParametricPlot[{t, 0}, {t, -6, 12}, Axes → None, PlotStyle → {Black, Thickness[0.007]}];

arq = ParametricPlot[{c1, c2, c3, c4}, {t, 0, Pi},
    Axes → {True, False}, PlotStyle → Directive[Orange, Opacity[0.6], Thickness[0.01]]];

Show[arq, suelo]
```



- Para representar el área común de ambos arcos se necesitan los puntos de intersección entre las circunferencias
  - $C_1 \cap C_3 :\ P_{13} = (2.9,\ 3.4409)$

```
NSolve[{x1 == x3, y1 == y3}, t]

{}
```

- **Nota.** `NSolve` devuelve una lista vacía ya que el sistema no tiene solución; el punto de intersección entre los grafos no se corresponde con el mismo valor de t en las ecuaciones paramétricas de las curvas

```
NSolve[{x^2 + y^2 == 4.5^2, (x - 5.8)^2 + y^2 == 4.5^2}, {x, y}]

{{x → 2.9, y → 3.44093}, {x → 2.9, y → -3.44093}}
```

- $C_1 \cap C_4:\ P_{14} = (1.7457, 4.1476)$

```
NSolve[{x^2 + y^2 == 4.5^2, (x - 5.8)^2 + y^2 == 5.8^2}, {x, y}]
{{x → 1.74569, y → 4.1476}, {x → 1.74569, y → -4.1476}}
```

- $C_2 \cap C_3:\ P_{23} = (4.0543, 4.1476)$

```
NSolve[{x^2 + y^2 == 5.8^2, (x - 5.8)^2 + y^2 == 4.5^2}, {x, y}]
{{x → 4.05431, y → 4.1476}, {x → 4.05431, y → -4.1476}}
```

- $C_2 \cap C_4:\ P_{24} = (2.9, 5.0230)$

```
NSolve[{x^2 + y^2 == 5.8^2, (x - 5.8)^2 + y^2 == 5.8^2}, {x, y}]
{{x → 2.9, y → 5.02295}, {x → 2.9, y → -5.02295}}
```

- Cálculo de los valores del parámetro t para los que quedan determinados los arcos de circunferencia que delimitan el área de intersección
  - circunferencia C_1
    - $P_{13} = (2.9, 3.4409)$

```
NSolve[x1 == 2.9, t]
{{t → -0.8705}, {t → 0.8705}}
```

    - $P_{14} = (1.7457, 4.1476)$

```
NSolve[x1 == 1.7457, t]
{{t → -1.17241}, {t → 1.17241}}
```

  - circunferencia C_2
    - $P_{23} = (4.0543, 4.1476)$

```
NSolve[x2 == 4.0543, t]
{{t → -0.796774}, {t → 0.796774}}
```

    - $P_{24} = (2.9, 5.0230)$

```
NSolve[x2 == 2.9, t]
{{t → -1.0472}, {t → 1.0472}}
```

  - circunferencia C_3
    - $P_{23} = (4.0543, 4.1476)$

```
NSolve[x3 == 4.0543, t]
{{t → -1.96918}, {t → 1.96918}}
```

    - $P_{13} = (2.9, 3.4409)$

```
NSolve[x3 == 2.9, t]
{{t → -2.27109}, {t → 2.27109}}
```

  - circunferencia C_4
    - $P_{24} = (2.9, 5.0230)$

```
NSolve[x4 == 2.9, t]
{{t → -2.0944}, {t → 2.0944}}
```

- $P_{14} = (1.7457, 4.1476)$

```
NSolve[x4 == 1.7457, t]
```

```
{{t → -2.34482}, {t → 2.34482}}
```

- Representación gráfica del área común con los puntos de intersección

```
g1 = ParametricPlot[c1, {t, 0.8705, 1.17241}, Axes → None, PlotStyle → {Thickness[0.01]}];
g2 = ParametricPlot[c2, {t, 0.7968, 1.0472}, Axes → None, PlotStyle → {Thickness[0.01]}];

g3 = ParametricPlot[c3, {t, 1.9692, 2.2711}, Axes → None, PlotStyle → {Thickness[0.01]}];
g4 = ParametricPlot[c4, {t, 2.0944, 2.3448}, Axes → None, PlotStyle → {Thickness[0.01]}];

v = {{2.9, 3.4409}, {1.7457, 4.1476}, {4.0543, 4.1476}, {2.9, 5.0230}};

p13 = Graphics[{Blue, PointSize[Large], Point[v[[1]]], Black, Text["P13", {v[[1, 1]], 3.3}]}];

p23 = Graphics[{Blue, PointSize[Large], Point[v[[3]]], Black, Text["P23", {v[[3, 1]], 4.0}]}];

p14 = Graphics[{Blue, PointSize[Large], Point[v[[2]]], Black, Text["P14", {v[[2, 1]], 4.0}]}];

p24 = Graphics[{Blue, PointSize[Large], Point[v[[4]]], Black, Text["P24", {v[[4, 1]], 5.15}]}];

area = Show[g1, g2, g3, g4, p13, p23, p14, p24, PlotRange → All]
```



Ejercicio nº2

Enunciado

Eero Saarinen fue un arquitecto norteamericano de origen finés. En 1947 diseñó el *Gateway Arch* para la ciudad de St. Louis en el estado de Missouri. Se trata de un impresionante monumento en forma de arco catenario cuya sección es un triángulo equilátero. Su construcción comenzó en 1963 y se completó en 1965. Constituye el arco más alto del mundo y, en su interior, contiene escaleras y un elevador para acceder a un mirador panorámico situado en la parte superior.

Imagen 2. Gateway Arch, St. Louis, Missouri

La imagen 2 es obra de Daniel Schwen.
Descargada de https://es.wikipedia.org/wiki/Arco_Gateway.

Represente gráficamente, en metros, los dos arcos catenarios que perfilan el monumento.

Debe tenerse en cuenta que, debido a la diferencia de anchura y altura entre el interior (intradós) y el exterior (extradós) del *Gateway Arch*, son necesarios dos arcos catenarios para poder representarlo con exactitud. La anchura de arco en la cota de suelo es de 630 pies en el extradós y de 536 pies en el intradós mientras que la altura máxima es de 630 y 612 pies en el extradós e intradós respectivamente.

La función *coseno hiperbólico*, $\cosh(k\,x)$, proporciona una magnifica herramienta para representar una catenaria si bien son necesarios algunos parámetros más para poder conferirle las dimensiones reales. Partiendo de $\cosh(0.01\,x)$ se puede establecer la siguiente ecuación para el *Gateway Arch* de St. Louis:

$$\boxed{y = a - b \cdot \cosh(0.01\,x)}$$

Resolución

```
Remove["Global`*"]
```

- Debido a la simetría de la figura, se establece un sistema cartesiano de referencia con origen en el eje de simetría a cota cero (suelo).
- Cambio en las unidades para resolver el ejercicio en metros donde $1\ pie = 0.3048\ \mathrm{m}$

```
factor = 0.3048
```

```
0.3048
```

  - la altura y anchura del extradós es la misma: 630 pies

```
hx = 630 * factor
```

```
192.024
```

  - altura y anchura del intradós

```
{hi = 612 * factor, ai = 536 * factor}
```

```
{186.538, 163.373}
```

- Para obtener las ecuaciones de las catenarias es necesario calcular los parámetros a y b algo que puede hacerse planteando, en cada caso, un sistema de dos ecuaciones sustituyendo las variables x e y por las coordenadas de dos puntos de la curva (el vértice y una de las intersecciones con el eje Ox, por ejemplo)
- Ecuación de la catenaria del extradós
  - puntos de intersección con el eje Ox (en pies): $A_1 = (-315, 0)$, $A_2 = (315, 0)$
  - vértice (en pies): $V_1 = (0, 630)$
  - obtención de los parámetros a y b (en metros)

```
sx = Solve[{0 == a - b Cosh[0.01 hx / 2], hx == a - b Cosh[0.01 × 0]}, {a, b}]
```

```
{{a → 578.057, b → 386.033}}
```

  - ecuación cartesiana: $y = 578.057 - 386.033 \cdot \cosh(0.01 \cdot x)$
  - ecuaciones paramétricas:

```
extra = {xax = t, yax = a - b * Cosh[0.01 xax] /. sx[[1]]}
```

```
{t, 578.057 - 386.033 Cosh[0.01 t]}
```

- variación del parámetro:

```
{0 - hx / 2, 0 + hx / 2}
```

```
{-96.012, 96.012}
```

$$A_x \equiv \begin{cases} x(t) = & t \\ y(t) = & 578.057 - 386.033 \cosh(0.01\, t) \end{cases} \quad \forall\, t \in [-96.012, 96.012]$$

- Ecuación de la catenaria del intradós
  - puntos de intersección con el eje Ox (en pies): $A_1 = (-268, 0)$, $A_2 = (268, 0)$
  - vértice (en pies): $V_1 = (0, 612)$
  - obtención de los parámetros a y b (en metros)

```
si = Solve[{0 == a - b Cosh[0.01 ai / 2], hi == a - b Cosh[0.01×0]}, {a, b}]
```

```
{{a → 715.568, b → 529.03}}
```

  - ecuación cartesiana: $y = 715.568 - 529.03 \cdot \cosh(0.01 \cdot x)$
  - ecuaciones paramétricas:

```
intra = {xai = t, yai = a - b * Cosh[0.01 xai] /. si[[1]]}
```

```
{t, 715.568 - 529.03 Cosh[0.01 t]}
```

```
{0 - ai / 2, 0 + ai / 2}
```

```
{-81.6864, 81.6864}
```

$$A_i \equiv \begin{cases} x(t) = & t \\ y(t) = & 715.568 - 529.03 \cosh(0.01\, t) \end{cases} \quad \forall\, t \in [-81.6864, 81.6864]$$

- Representación gráfica de las catenarias que perfilan el monumento
  - extradós

```
extradós = ParametricPlot[extra, {t, -96.012, 96.012}, PlotStyle → Directive[Thick, Gray]];
```

  - intradós

```
intradós = ParametricPlot[intra, {t, -81.6864, 81.6864}, PlotStyle → Directive[Thick, Gray]];
```

  - juntando ambas

```
suelo = ParametricPlot[{t, 0}, {t, -100, 100}, Axes → None, PlotStyle → {Black, Thickness[0.007]}];
```

```
Show[extradós, intradós, suelo]
```





Imagen 3. Simulación del arco junto a Saarinen con una maqueta del Gateway Arch

La fotografía de Saarinen junto a la maqueta mostrada en la imagen 3 ha sido descargada de:

https://www.wikiwand.com/en/Gateway_Arch

Ejercicio nº3

Enunciado

La Torre de Collserola es una esbelta torre de comunicaciones erigida en Barcelona con motivo de los Juegos Olímpicos por el arquitecto británico Norman Foster y la ingeniería Ove Arup. La torre se eleva 266 metros sobre el nivel del suelo a los que hay que añadir los 445 del terreno, resultando una de las estructuras más visibles desde toda la ciudad.

Imagen 4. Torre de Collserola, Barcelona

La imagen 4 es de dominio público.
Descargada de https://es.m.wikipedia.org/wiki/Archivo:Torre_Collserola,_Norman_Foster.jpg.

La torre consta de un fuste hueco de hormigón de tan solo 4.5 metros de diámetro que soporta en su parte superior un sistema de tres tirantes y se ancla al suelo mediante otro sistema de tirantes. El peso total de la estructura metálica es de 3.000 toneladas.

La Torre de Collserola cuenta con 12 plataformas entre los 84 y los 136 metros, cada una de las cuales tiene una geometría característica denominada Triángulo de Reuleaux. Consiste en la intersección de tres circunferencias cuyo centro se sitúa en cada uno de los vértices de un triángulo equilátero coincidiendo el radio de cada una con la longitud de los lados del triángulo. Debe su nombre al ingeniero alemán Franz Reuleaux.



Imagen 5. Generación de un triángulo Reuleaux (imagen propia)



Imagen 6. Planta en plano

Imagen 6 descargada de:

https://es.wikiarquitectura.com/edificio/torre-de-collserola/#torre-collserola-pl

Sabiendo que el área de cada una de las plantas es de 429 metros cuadrados, determine el diámetro de las circunferencias que dan forma a la planta.

Halle las ecuaciones paramétricas que definen el contorno de cada planta. Represente gráficamente el triángulo de Reuleaux resultante.

Resolución

```
Remove["Global`*"]
```

- Debido a la simetría de la figura, se establece el sistema cartesiano de referencia mostrado en la siguiente imagen:



Imagen 7. Sistema de referencia empleado (imagen propia)

- Ecuaciones cartesianas de las curvas suponiendo que la longitud del lado del triángulo equilátero (radio) es r metros
    - Circunferencia 1, C_1, centrada en el punto $V_1 = \left(-\frac{r}{2},\, 0\right)$:

$$C_1 \equiv \left(x+\tfrac{r}{2}\right)^2 + y^2 = r^2 \quad \forall\, x \in \left[-\frac{3r}{2},\, \frac{r}{2}\right]$$

    - Circunferencia 2, C_2, centrada en el punto $V_2 = \left(0,\, \frac{\sqrt{3}\, r}{2}\right)$ debido a la relación entre la altura h y el lado r del triángulo equilátero, $h^2 + \left(\frac{r}{2}\right)^2 = r^2 \Rightarrow h = \frac{\sqrt{3}}{2} r$:

$$C_2 \equiv x^2 + \left(y - \frac{\sqrt{3}\, r}{2}\right)^2 = r^2 \quad \forall\, x \in \left[-\frac{r}{2},\, \frac{r}{2}\right]$$

    - Circunferencia 3, C_3, centrada en el punto $V_3 = \left(\frac{r}{2},\, 0\right)$:

$$C_3 \equiv \left(x-\tfrac{r}{2}\right)^2 + y^2 = r^2 \quad \forall\, x \in \left[-\frac{r}{2},\, \frac{3r}{2}\right]$$

- Para determinar el diámetro pedido es necesario calcular el radio a partir del dato facilitado, el área de la planta
- Debido a la simetría de la figura, se considera la mitad del dominio limitado:
    - superiormente por el arco de C_3 : $y = \sqrt{r^2 - \left(x-\frac{r}{2}\right)^2} \quad \forall\, x \in \left[-\frac{r}{2},\, 0\right]$

```
Solve[(x - r/2)^2 + y^2 == r^2, y]
```

$$\left\{\left\{y \to -\frac{1}{2}\sqrt{3 r^2 + 4 r x - 4 x^2}\right\},\ \left\{y \to \frac{1}{2}\sqrt{3 r^2 + 4 r x - 4 x^2}\right\}\right\}$$

    - inferiormente por el arco de C_2 : $y = \frac{\sqrt{3}\, r}{2} - \sqrt{r^2 - x^2} \quad \forall\, x \in \left[-\frac{r}{2},\, 0\right]$

```
Solve[x^2 + (y - Sqrt[3] r/2)^2 == r^2, y] // Expand
```

$$\left\{\left\{y \to \frac{\sqrt{3}\, r}{2} - \sqrt{r^2 - x^2}\right\},\ \left\{y \to \frac{\sqrt{3}\, r}{2} + \sqrt{r^2 - x^2}\right\}\right\}$$



Imagen 8. Triángulo de Reuleaux (imagen propia)

- Se recuerda la fórmula para calcular el área del dominio limitado entre las gráficas de dos funciones $y = f(x)$, $y = g(x)$ y las rectas verticales $x = a$, $x = b$:



$$\boxed{Ar = \int_a^b [f(x) - g(x)]\, dx}$$

Imagen 9. Dominio simbólico y fórmula de cálculo del área (imagen propia)

- Se calcula la integral extendida a la mitad descrita del dominio:

$$\text{Integrate}\left[\frac{1}{2}\sqrt{3\,r^2 + 4\,r\,x - 4\,x^2} - \left(\frac{\sqrt{3}\,r}{2} - \sqrt{r^2 - x^2}\right),\ \{x,\ -r/2,\ 0\}\right]$$

$$\frac{1}{4}\,r\left(-\sqrt{3}\,r + \pi\sqrt{r^2}\right)$$

- Como el área de la planta es de 429 m^2, puede calcularse el radio

$$s = \text{NSolve}\left[\frac{1}{4}\,r\left(-\sqrt{3}\,r + \pi\sqrt{r^2}\right) == 429/2,\ r\right]$$

```
{{r → 24.672}, {r → 0. - 13.2683 i}}
```

```
diámetro = 2 r /. s[[1]]
```

```
49.344
```

- Por tanto, el diámetro de las circunferencias que generan el triángulo de Reuleaux es $\boxed{d = 49.344\, m}$
- Ecuaciones paramétricas de las curvas:

$$\left\{r/2,\ \frac{\sqrt{3}\,r}{2}\right\} \text{ /. s[[1]] (* calculado para la parametrización de las líneas *)}$$

```
{12.336, 21.3666}
```

  - Circunferencia 1, C_1: $\boxed{C_1 \equiv \begin{cases} x(t) = & -12.336 + 24.672\cos(t) \\ y(t) = & 24.672\,\text{sen}(t) \end{cases} \quad \forall\, t \in [0,\, 2\pi]}$
  - Circunferencia 2, C_2: $\boxed{C_2 \equiv \begin{cases} x(t) = & 24.672\cos(t) \\ y(t) = & 21.367 + 24.672\,\text{sen}(t) \end{cases} \quad \forall\, t \in [0,\, 2\pi]}$
  - Circunferencia 3, C_3: $\boxed{C_3 \equiv \begin{cases} x(t) = & 12.336 + 24.672\cos(t) \\ y(t) = & 24.672\,\text{sen}(t) \end{cases} \quad \forall\, t \in [0,\, 2\pi]}$

- Representación gráfica

```
r = 24.672;
```

- definición de ecuaciones

```
c1 = {x1[t_] = -r / 2 + r Cos[t], y1[t_] = r Sin[t]};

c2 = {x2[t_] = r Cos[t], y2[t_] = Sqrt[3] r / 2 + r Sin[t]};

c3 = {x3[t_] = r / 2 + r Cos[t], y3[t_] = r Sin[t]};
```

- trazado de las circunferencias

```
ParametricPlot[{c1, c2, c3}, {t, 0, 2 Pi}]
```



- trazado de los arcos

```
Reuleaux =
 ParametricPlot[{{x1[t], y1[t]}, {x3[t + 2 Pi / 3], y3[t + 2 Pi / 3]}, {x2[t + 4 / 3 Pi], y2[t + 4 / 3 Pi]}},
  {t, 0, Pi / 3}, PlotStyle → Directive[Thick, Gray]]
```



Ejercicio nº4

Enunciado

El Cementerio de la familia Brion es la obra maestra del arquitecto italiano Caro Scarpa. Se construyó cerca de Treviso, entre 1969 y 1978, como anexo a un cementerio municipal. Esta tumba familiar está repleta de simbolismo sobre la vida y la muerte, como la intersección de dos círculos a modo de *vesica piscis*.

La forma *vesica piscis* canónica se caracteriza porque cada circunferencia pasa por el centro de la otra. En este caso no es así y las circunferencias están desplazadas arbitrariamente formando una peculiar intersección de 120.5 centímetros de altura en su parte central.



Imagen 10. Cementerio Brion, Treviso

La imagen 10 es de dominio público.
Descargada de https://upload.wikimedia.org/wikipedia/commons/c/ca/Tomba_Brion_039.jpg.

Halle y represente las ecuaciones paramétricas de ambas circunferencias sabiendo que arrancan sobre un zócalo de 15 centímetros sobre el suelo y su diámetro es de 170 centímetros.



Imagen 11. Generación de la *vesica piscis* (imagen propia)

Determine, también, la altura en la parte central si la *vesica piscis* generada fuera canónica.

Resolución

```
Remove["Global`*"]
```

- En primer lugar, se establece un sistema cartesiano de referencia tal que el eje de abscisas se establece en la cota de suelo y el de ordenadas se hace coincidir con el eje de simetría de la figura.
- Si se trazara una *vesica piscis* canónica con las dos circunferencias de radio $R = 85\,cm$ se tendría (sumando 15 centímetros a la ordenada de los centros tal como se indica en el enunciado):
  - circunferencia C_1 centrada en $A = \left(-\frac{R}{2}, R+15\right) = (-42.5, 100)$
  - circunferencia C_2 centrada en $B = \left(\frac{R}{2}, R+15\right) = (42.5, 100)$
- Ecuaciones implícitas de ambas circunferencias:

$$C_1 \equiv (x+42.5)^2 + (y-100)^2 = 85^2 \quad \forall\, x \in [-127.5, 42.5]$$

$$C_2 \equiv (x-42.5)^2 + (y-100)^2 = 85^2 \quad \forall\, x \in [-127.5, 42.5]$$

- Parametrización de las circunferencias:

$$C_1 \equiv \begin{cases} x(t) = & -42.5 + 85\cos(t) \\ y(t) = & 100 + 85\,\mathrm{sen}(t) \end{cases} \quad \forall\, t \in [0, 2\pi] \quad C_2 \equiv \begin{cases} x(t) = & 42.5 + 85\cos(t) \\ y(t) = & 100 + 85\,\mathrm{sen}(t) \end{cases} \quad \forall\, t \in [0, 2\pi]$$

```
{r = 170 / 2, a = r / 2}; (* la variable a denota la distancia de los centros al eje de ordenadas*)
```

```
c1 = {x1 = -a + r Cos[t], y1 = r + 15 + r Sin[t]};
c2 = {x2 = a + r Cos[t], y2 = r + 15 + r Sin[t]};
```

- Representación gráfica:

```
vesica1 = ParametricPlot[{c1, c2}, {t, 0, 2 Pi}];

g1 = Graphics[{Black, Text["C1", {-115, 100}]}];
g2 = Graphics[{Black, Text["C2", {115, 100}]}];

Show[vesica1, g1, g2]
```



- La *vesica piscis* está comprendida entre los puntos de intersección de las circunferencias
- Cálculo del parámetro t en los puntos de intersección, $C_1 \cap C_2$ (cada ecuación admite infinitas soluciones):

```
Solve[x1 == 0, t]
```

$$\left\{\left\{t \to \text{ConditionalExpression}\left[-\text{ArcCos}\left[\frac{a}{10}\right] + 2\pi C[1],\ C[1] \in \mathbb{Z}\right]\right\},\ \left\{t \to \text{ConditionalExpression}\left[\text{ArcCos}\left[\frac{a}{10}\right] + 2\pi C[1],\ C[1] \in \mathbb{Z}\right]\right\}\right\}$$

```
Solve[x2 == 0, t]
```

$$\left\{\left\{t \to \text{ConditionalExpression}\left[-\text{ArcCos}\left[-\frac{a}{10}\right] + 2\pi C[1],\ C[1] \in \mathbb{Z}\right]\right\},\ \left\{t \to \text{ConditionalExpression}\left[\text{ArcCos}\left[-\frac{a}{10}\right] + 2\pi C[1],\ C[1] \in \mathbb{Z}\right]\right\}\right\}$$

- Cálculo de la altura de la *vesica piscis* restando las ordenadas de los puntos de intersección:

$$\left(y1\ /.\ t \to \frac{\pi}{3}\right) - \left(y1\ /.\ t \to -\frac{\pi}{3}\right)\ //\ N$$

```
147.224
```

- En el caso de dos circunferencias de radio $R = 85\,cm$ la *vesica piscis* canónica tiene, por tanto, una altura $\boxed{h = 147.224\,cm}$.
- Para la segunda parte del ejercicio debe calcularse la traslación horizontal de las circunferencias con respecto a la *vesica piscis* canónica.
- Se recuerda que la variable **a** se utiliza para contener la distancia de los centros de las circunferencias al eje de ordenadas (valor absoluto de las abscisas); es el valor que debe determinarse a partir de los datos conocidos.

```
Clear[a]
```

- Parametrización de las circunferencias:

$$C_1 \equiv \begin{cases} x(t) = & -a + 85\cos(t) \\ y(t) = & 100 + 85\,\text{sen}(t) \end{cases} \quad \forall\, t \in [0, 2\pi] \qquad C_2 \equiv \begin{cases} x(t) = & a + 85\cos(t) \\ y(t) = & 100 + 85\,\text{sen}(t) \end{cases} \quad \forall\, t \in [0, 2\pi]$$

```
c1 = {x1 = -a + r Cos[t], y1 = r + 15 + r Sin[t]};
c2 = {x2 = a + r Cos[t], y2 = r + 15 + r Sin[t]};
```

- Se repite el proceso realizado anteriormente para calcular la altura de la *vesica piscis*; al ser conocida en este caso (120.5 centímetros), la incógnita que se calcula es la abscisa de los centros de las circunferencias
  - cálculo del parámetro t en los puntos de intersección, $C_1 \cap C_2$:

```
Solve[x1 == 0, t]
```

$$\left\{\left\{t\to \text{ConditionalExpression}\left[-\text{ArcCos}\left[\frac{a}{10}\right]+2\pi C[1],\ C[1]\in\mathbb{Z}\right]\right\},\left\{t\to \text{ConditionalExpression}\left[\text{ArcCos}\left[\frac{a}{10}\right]+2\pi C[1],\ C[1]\in\mathbb{Z}\right]\right\}\right\}$$

```
Solve[x2 == 0, t]
```

$$\left\{\left\{t\to \text{ConditionalExpression}\left[-\text{ArcCos}\left[-\frac{a}{10}\right]+2\pi C[1],\ C[1]\in\mathbb{Z}\right]\right\},\left\{t\to \text{ConditionalExpression}\left[\text{ArcCos}\left[-\frac{a}{10}\right]+2\pi C[1],\ C[1]\in\mathbb{Z}\right]\right\}\right\}$$

  - restando las ordenadas de los puntos de intersección:

$$\left(\text{y1 /. t -> ArcCos}\left[\frac{a}{85}\right]\right)-\left(\text{y1 /. t}\to -\text{ArcCos}\left[\frac{a}{85}\right]\right)\ \text{// N}$$

$$170.\sqrt{1.-0.000138408\,a^2}$$

  - cálculo de la incógnita **a** sabiendo que $h = 120.5\,cm$:

$$\text{Solve}\left[170\sqrt{1-0.00013924\,a^2}==120.5,\ a\right]$$

$$\{\{a\to -59.7785\},\{a\to 59.7785\}\}$$

- Entonces, la parametrización y representación gráfica de las circunferencias resultantes es:

$C_1 \equiv \begin{cases} x(t) = & -59.78 + 85\cos(t) \\ y(t) = & 100 + 85\,\text{sen}(t) \end{cases} \quad \forall\, t \in [0, 2\pi]$	$C_2 \equiv \begin{cases} x(t) = & 59.78 + 85\cos(t) \\ y(t) = & 100 + 85\,\text{sen}(t) \end{cases} \quad \forall\, t \in [0, 2\pi]$

```
{r = 85, a = 59.78};

ParametricPlot[{c1, c2}, {t, 0, 2 Pi}]
```




```
vesica2 = ParametricPlot[{c1, c2}, {t, 0, 2 Pi},
   PlotStyle → {Directive[Thickness[0.02], Brown], Directive[Thickness[0.02], Blue]}];

suelo = ParametricPlot[{t, 0}, {t, -175, 175}, Axes → None, PlotStyle → {Gray, Thickness[0.02]}];
```

```
fondo = RegionPlot[(x + 59.78)^2 + (y - 100)^2 > 87^2 && (x - 59.78)^2 + (y - 100)^2 > 87^2,
   {x, -175, 175}, {y, 4, 200}, PlotStyle → {Opacity[0.6], LightGray},
   PlotRange -> All, AspectRatio → Automatic, BoundaryStyle → None];
```

- la función **RegionPlot** permite representar la región para la que el primer argumento se evalúa como Verdadero, aquí se ha utilizado de forma estética para simular la pared de hormigón

```
Show[vesica2, suelo, fondo, Axes → None]
```



Ejercicio nº5

Enunciado

Marina City es un complejo de edificios diseñado por el arquitecto Bertrand Goldberg para la ciudad de Chicago, en el estado de Illinois. Se construyó en el año 1964 en la orilla norte del río Chicago en el centro de la ciudad. En el complejo se encuentran dos torres con una característica forma de mazorca de maíz que, en el momento de su inauguración, fueron los edificios residenciales y las estructuras de hormigón más altas del mundo (179 metros).

Imagen 12. Marina City, Chicago, Illinois

La imagen 12 es obra de Spikebrennan.
Descargada de https://es.wikipedia.org/wiki/Marina_City#/media/Archivo:Marina_towers.JPG.

Las primeras 19 plantas, destinadas a aparcamiento, están diseñadas a partir de una rampa helicoidal al aire libre. Le siguen 40 plantas de apartamentos dispuestas alrededor de un núcleo de hormigón armado de 35 pies de diámetro en los que predominan las formas curvas, sin prácticamente ángulos rectos en su interior. La superficie residencial construida se distribuye en cada piso en 16 unidades habitacionales, cada una de las cuales queda comprendida entre un arco de la epicicloide y el núcleo central.

Imagen 13. Detalle de la fachada (imagen propia)

En la cota en planta, la forma de mazorca de maíz se traduce matemáticamente en una curva epicicloide siendo el diámetro de su circunferencia directriz de 120 pies.



Imagen 14. Cota en planta y curva epicicloide (imagen propia)

Fórmula de una curva epicicloide siendo $r = b$ el radio de la circunferencia generatriz y $R = a$ el radio de la directriz tal como se ilustra en la imagen:

$$\boxed{C \equiv \begin{cases} x(t) = & (a+b)\cos(t) - b\cdot\cos\left(\frac{a+b}{b}\cdot t\right) \\ y(t) = & (a+b)\,\text{sen}(t) - b\cdot\text{sen}\left(\frac{a+b}{b}\cdot t\right) \end{cases} \quad \forall\, t \in [0, 2\pi]}$$

A partir del diagrama de las cotas en planta y la fórmula de la epicicloide, calcule:

- los radios de las circunferencias que generan la epicicloide que da forma a las torres
- el perímetro, en metros, de la fachada en una sección transversal
- el área de cada unidad habitacional, en metros cuadrados

Se facilitan las fórmulas para el cálculo de la longitud y el área del dominio limitado por un arco de curva dada en forma paramétrica:

$$\boxed{L = \int_\alpha^\beta \sqrt{[x'(t)]^2 + [y'(t)]^2}\; dt} \qquad \boxed{Ar = -\int_\alpha^\beta y(t)\cdot x'(t)\, dt}$$

Resolución

```
Remove["Global`*"]
```

- Se establece un sistema cartesiano de referencia con origen en el centro de la circunferencia directriz

- El radio de la circunferencia directriz, C_d, es: $\boxed{R = a = 60\ pies}$

$$C_d \equiv x^2 + y^2 = 60^2 \quad \forall\ x \in [-60, 60]$$

- La razón entre los radios $k = \frac{R}{r}$ se corresponde con el número de arcos de la epicicloide; en este caso, el radio de la circunferencia generatriz es: $k = 16 \Rightarrow r = \frac{R}{k} = \frac{60}{16} \Rightarrow \boxed{r = b = \frac{15}{4}\ pies}$

- Con lo que ya se puede parametrizar y representar la epicicloide:

```
{a = 120 / 2; b = a / 16};

epic =
 {xe[t_] = (a + b) Cos[t] - b * Cos[t (a + b) / b], ye[t_] = (a + b) Sin[t] - b * Sin[t (a + b) / b]} // Simplify
```

$$\left\{\frac{15}{4} (17 \operatorname{Cos}[t] - \operatorname{Cos}[17 t]), \frac{15}{4} (17 \operatorname{Sin}[t] - \operatorname{Sin}[17 t])\right\}$$

$$\boxed{C \equiv \begin{cases} x(t) = \frac{15}{4} \cdot [17 \cos(t) - \cos(17 t)] \\ y(t) = \frac{15}{4} \cdot [17 \operatorname{sen}(t) - \operatorname{sen}(17 t)] \end{cases} \quad \forall\ t \in [0, 2\pi]}$$

```
cd = {xc = a * Cos[t], yc = a * Sin[t]};

rn = 35 / 2; (* radio del núcleo central *)

cn = {xc = rn * Cos[t], yc = rn * Sin[t]};

planta = ParametricPlot[{epic, cd, cn}, {t, 0, 2 Pi}, PlotStyle → {Directive[Thickness[0.01], Red],
    Directive[Thickness[0.005], Blue], Directive[Thickness[0.005], Blue]}]
```



- Cambio en las unidades para resolver el ejercicio en metros: $1\ pie = 0.3048\,\mathrm{m}$

```
factor = 0.3048;
```

- perímetro en metros de la fachada en una sección transversal: $\boxed{p = 155.45\ m}$

```
perimetro = factor * NIntegrate[Sqrt[xe'[t]^2 + ye'[t]^2], {t, 0, 2 Pi}]
```

155.448

- el área de cada unidad habitacional, en metros cuadrados: $\boxed{Ar = 72.91\ m^2}$

$$\texttt{area} = \frac{\texttt{factor}^2}{16} \left(-\texttt{NIntegrate[ye[t] * xe'[t], \{t, 0, 2 Pi\}]} - \pi * \texttt{rn}^2\right)$$

72.9089